

Example 3: Time-Invariant Predictors of Practice Effects (uses same data as Example 2) (complete data, syntax, and output available for SAS, SPSS, and STATA electronically)

In this example we will examine time-invariant predictors of individual differences in intercepts, linear slopes, and quadratic slopes representing improvement in RT (in msec) across six practice sessions. We will examine age, abstract reasoning, and education in sequential conditional (predictor) polynomial models.

SAS Code for Data Manipulation:

```
* Centering level-2 predictor variables for analysis;
DATA work.example23; SET work.example23;
    age80 = age - 80;          * Convenient value;
    reas22 = absreas - 22;     * Near sample mean;
    LABEL age80 = "age80: Age Centered (0=80)"
           reas22 = "reas22: Abstract Reasoning Centered (0=22)";
    * Make education a grouping variable for purpose of demonstration only;
    IF educyrs LE 12          THEN educgrp=1;
    ELSE IF educyrs GT 12 AND EducYrs LE 16 THEN educgrp=2;
    ELSE IF educyrs GT 16          THEN educgrp=3;
    ELSE IF educyrs = .          THEN educgrp=.;
    LABEL educgrp = "educgrp: Education Group (1=HS, 2=BA, 3=GRAD)";
* Removing cases with missing predictors;
IF NMISS(age80, reas22, educgrp)>0 THEN DELETE;
RUN;
```

SPSS Code for Data Manipulation:

```
* Centering level-2 predictor variables for analysis.
DATASET ACTIVATED example23 WINDOW=FRONT.
COMPUTE age80 = age - 80.
COMPUTE reas22 = absreas - 22.
VARIABLE LABELS
    age80 "age80: Age Centered (0=80)"
    reas22 "reas22: Abstract Reasoning Centered (0=22)".
* Make education a grouping variable for purpose of demonstration only.
IF educyrs LE 12          educgrp=1.
IF educyrs GT 12 AND educyrs LE 16 educgrp=2.
IF educyrs GT 16          educgrp=3.
VARIABLE LABELS educgrp "educgrp: Education Group (1=HS, 2=BA, 3=GRAD)".
* Removing cases with missing predictors.
SELECT IF (NVALID(age80, reas22, educgrp)=3).
EXECUTE.
```

STATA Code for Data Manipulation:

```
* centering level-2 predictor variables for analysis
gen age80 = age - 80
gen reas22 = absreas - 22
label variable age80 "age80: Age Centered (0=80 years)"
label variable reas22 "reas22: Abstract Reasoning Centered (0=22)"
* make education a grouping variable for purpose of demonstration only
gen educgrp=.
replace educgrp=1 if (educyrs <= 12)
replace educgrp=2 if (educyrs > 12 & educyrs <= 16)
replace educgrp=3 if (educyrs > 16)
label variable educgrp "educgrp: Education Group (1=HS, 2=BA, 3=GRAD)"

* create new variable to hold number of missing cases
* then drop cases with incomplete predictors
egen nummiss = rowmiss(age80 reas22 educgrp)
drop if nummiss>0
```

Model 3b. Random Quadratic Time Baseline (in ML now)

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + U_{2i}$$

```
TITLE "SAS Model 3b: Random Quadratic Time Baseline in ML";
PROC MIXED DATA=work.example23 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess / SOLUTION DDFM=Satterthwaite OUTPM=work.TimePred;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G GCORR V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN;
PROC CORR NOSIMPLE DATA=work.TimePred; VAR nm3rt pred; RUN;
```

The OUTPM in SAS, /SAVE in SPSS, and predict in STATA calculate outcomes predicted by the fixed effects. We can then correlate the predicted and actual outcomes to get total R^2 (actual variance explained).

```
TITLE "SPSS Model 3b: Random Quadratic Time Baseline in ML".
MIXED nm3rt BY ID session WITH c1sess
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess c1sess*c1sess
  /RANDOM = INTERCEPT c1sess c1sess*c1sess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (predtime).
CORRELATIONS nm3rt predtime.
```

```
* STATA Model 3b: Random Quadratic Time Baseline in ML
xtmixed nm3rt c.c1sess c.c1sess#c.c1sess, || id: c1sess c1sess2, ///
  variance mle covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store Baseline, // save LL for LRT
  predict predtime // save fixed-effect predicted outcomes
corr nm3rt predtime // get total r to make r2
```

SAS Output:

Estimated G Matrix

ID:

Participant

Row	Effect	ID	Col1	Col2	Col3
1	Intercept	101	273306	-35262	3845.38
2	c1sess	101	-35262	25438	-3837.76
3	c1sess*c1sess	101	3845.38	-3837.76	622.81

Estimated G Correlation Matrix

ID:

Participant

Row	Effect	ID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4229	0.2947
2	c1sess	101	-0.4229	1.0000	-0.9642
3	c1sess*c1sess	101	0.2947	-0.9642	1.0000

Note how correlated the linear and quadratic random slopes are...

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	ID	273306	40828	6.69	<.0001	Random Intercept variance
UN(2,1)	ID	-35262	11765	-3.00	0.0027	Intercept-Linear slope covariance
UN(2,2)	ID	25438	5781.19	4.40	<.0001	Random Linear slope variance
UN(3,1)	ID	3845.38	1920.35	2.00	0.0452	Intercept-Quadratic slope covariance
UN(3,2)	ID	-3837.76	968.79	-3.96	<.0001	Linear-Quadratic slope covariance
UN(3,3)	ID	622.81	169.99	3.66	0.0001	Random Quadratic slope variance
session	ID	20298	1649.11	12.31	<.0001	Residual variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8321.8	10	8341.8	8342.1	8352.4	8367.9	8377.9

In ML, the #parms is ALL parms (both sides of model).

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1945.85	53.5825	101	36.32	<.0001
c1sess	-120.90	19.9481	101	-6.06	<.0001
c1sess*c1sess	13.8656	3.3985	101	4.08	<.0001

Pearson Correlation Coefficients, N = 606

Prob > |r| under H0: Rho=0

	nm3rt	Pred
nm3rt	1.00000	0.19167
nm3rt: Number-Match 3 RT in ms		<.0001

R = .1917, so R² for time = .0367 The model for the means (fixed linear and quadratic session effects so far) accounted for ~4% of the variance in RT.

Model 4a. Age as Predictor of Intercept, Linear, and Quadratic Time Slopes

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + U_{2i}$$

```
TITLE1 "SAS Model 4a: Age as Predictor of Intercept, Linear, and Quadratic";
PROC MIXED DATA=work.example23 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = c1sess c1sess*c1sess age80 c1sess*age80 c1sess*c1sess*age80
    / SOLUTION DDFM=Satterthwaite OUTPM=work.AgePred;
  RANDOM INTERCEPT c1sess c1sess*c1sess / G GCORR TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  * Requesting additional effects for age;
  ESTIMATE "Age Effect at Session 1" age80 1 c1sess*age80 0 c1sess*c1sess*age80 0;
  ESTIMATE "Age Effect at Session 2" age80 1 c1sess*age80 1 c1sess*c1sess*age80 1;
  ESTIMATE "Age Effect at Session 3" age80 1 c1sess*age80 2 c1sess*c1sess*age80 4;
  ESTIMATE "Age Effect at Session 4" age80 1 c1sess*age80 3 c1sess*c1sess*age80 9;
  ESTIMATE "Age Effect at Session 5" age80 1 c1sess*age80 4 c1sess*c1sess*age80 16;
  ESTIMATE "Age Effect at Session 6" age80 1 c1sess*age80 5 c1sess*c1sess*age80 25;
RUN; PROC CORR NOSIMPLE DATA=work.AgePred; VAR nm3rt pred; RUN;
```

```
TITLE "SPSS Model 4a: Age as Predictor of Intercept, Linear, and Quadratic".
MIXED nm3rt BY ID session WITH c1sess age80
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = c1sess c1sess*c1sess age80 c1sess*age80 c1sess*c1sess*age80
```

```

/RANDOM = INTERCEPT c1sess c1sess*c1sess | SUBJECT(ID) COVTYPE(UN)
/REPEATED = session | SUBJECT(ID) COVTYPE(ID)
/SAVE = FIXPRED (predage)
/TEST = "Age Effect at Session 1" age80 1 c1sess*age80 0 c1sess*c1sess*age80 0
/TEST = "Age Effect at Session 2" age80 1 c1sess*age80 1 c1sess*c1sess*age80 1
/TEST = "Age Effect at Session 3" age80 1 c1sess*age80 2 c1sess*c1sess*age80 4
/TEST = "Age Effect at Session 4" age80 1 c1sess*age80 3 c1sess*c1sess*age80 9
/TEST = "Age Effect at Session 5" age80 1 c1sess*age80 4 c1sess*c1sess*age80 16
/TEST = "Age Effect at Session 6" age80 1 c1sess*age80 5 c1sess*c1sess*age80 25.
CORRELATIONS nm3rt predage.

* STATA Model 4a: Age as Predictor of Intercept, Linear, and Quadratic
xtmixed nm3rt c.c1sess c.c1sess#c.c1sess ///
  c.age80 c.age80#c.c1sess c.age80#c.c1sess#c.c1sess, ///
  || id: c1sess c1sess2, ///
  variance ml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store age, // save LL for LRT
  lrtest Age Baseline, // LRT against non-age baseline
  predict predage // save fixed-effect predicted outcomes
  margins, at(c.c1sess=(0(1)5)) dydx(c.age80) vsquish // age slope per session
  margins, at(c.c1sess=(0(1)5) c.age80=(-5 0 5)) vsquish // predictions per session
  marginsplot, name(predicted_age, replace) // plot age predictions
corr nm3rt predage // get total r to make r2

```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	242456	36490	6.64	<.0001 → intercept variance reduced by 11.29%
UN(2,1)	ID	-29320	10863	-2.70	0.0070
UN(2,2)	ID	24294	5623.77	4.32	<.0001 → linear slope variance reduced by 4.50%
UN(3,1)	ID	3132.88	1792.88	1.75	0.0806
UN(3,2)	ID	-3700.54	949.39	-3.90	<.0001
UN(3,3)	ID	606.35	167.75	3.61	0.0002 → quad slope variance reduced by 2.64%
session	ID	20298	1649.11	12.31	<.0001 → residual variance not reduced

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8310.2	13	8336.2	8336.8	8350.0	8370.2	8383.2

Is the age model (4a) better than the baseline random quadratic model (3b)?

Yes, $-2\Delta LL = 11.6$ on $df=3$, $p=.009$

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1950.69	50.6713	101	38.50	<.0001
c1sess	-121.83	19.6695	101	-6.19	<.0001
c1sess*c1sess	13.9774	3.3757	101	4.14	<.0001
age80	29.0495	8.3774	101	3.47	0.0008
c1sess*age80	-5.5946	3.2519	101	-1.72	0.0884
c1sess*c1sess*age80	0.6709	0.5581	101	1.20	0.2321

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Age Effect at Session 1	29.0495	8.3774	101	3.47	0.0008
Age Effect at Session 2	24.1258	7.6097	101	3.17	0.0020
Age Effect at Session 3	20.5439	7.4593	101	2.75	0.0070
Age Effect at Session 4	18.3038	7.3302	101	2.50	0.0141
Age Effect at Session 5	17.4056	7.0715	101	2.46	0.0155
Age Effect at Session 6	17.8492	7.0545	101	2.53	0.0129

These are the simple slopes for the effect of age per session.


Pearson Correlation Coefficients, N = 606
 Prob > |r| under H0: Rho=0

	nm3rt	Pred
nm3rt	1.00000	0.32688
nm3rt: Number-Match 3 RT in ms		<.0001

$R = .32689$, so R^2 for time+age = .1069

The fixed effects of time before accounted for ~3.7% of the variance in RT, so there is a net increase of ~7% due to age.

The pattern of the interaction is shown by the simple effects of age at each session, graphed below.



Model 5a. +Abstract Reasoning as Predictor of Intercept, Linear, and Quadratic Time Slopes

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + U_{2i}$$

```

TITLE "SAS Model 5a: +Reasoning as Predictor of Intercept, Linear, and Quadratic";
PROC MIXED DATA=work.example23 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
               reas22 clsess*reas22 clsess*clsess*reas22
               / SOLUTION DDFM=Satterthwaite OUTPM=work.ReasPred;
  RANDOM INTERCEPT clsess clsess*clsess / G GCORR TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
RUN; PROC CORR NOSIMPLE DATA=work.ReasPred; VAR nm3rt pred; RUN;
TITLE "SPSS Model 5a: +Reasoning as Predictor of Intercept, Linear, and Quadratic".
MIXED nm3rt BY ID session WITH clsess age80 reas22
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
           reas22 clsess*reas22 clsess*clsess*reas22
  /RANDOM = INTERCEPT clsess clsess*clsess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (predreas).
CORRELATIONS nm3rt predreas

```

```

* STATA Model 5a: +Reasoning as Predictor of Intercept, Linear, and Quadratic
xtmixed nm3rt c.c1sess c.c1sess#c.c1sess ///
      c.age80 c.age80#c.c1sess c.age80#c.c1sess#c.c1sess ///
      c.reas22 c.reas22#c.c1sess c.reas22#c.c1sess#c.c1sess, ///
      || id: c1sess c1sess2, ///
      variance ml covariance(un) residuals(independent,t(session)),
      estat ic, n(101),
      estat recovariance, level(id),
      estimates store Reas, // save LL for LRT
      lrtest Reas Age, // LRT against age baseline
      predict predreas // save fixed-effect predicted outcomes
corr nm3rt predreas // get total r to make r2

```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	228049	34464	6.62	<.0001 → intercept variance reduced by 5.94%
UN(2,1)	ID	-31230	10649	-2.93	0.0034
UN(2,2)	ID	24041	5588.98	4.30	<.0001 → linear slope variance reduced by 1.04%
UN(3,1)	ID	3748.22	1746.63	2.15	0.0319
UN(3,2)	ID	-3618.98	937.05	-3.86	0.0001
UN(3,3)	ID	580.07	164.19	3.53	0.0002 → quad slope variance reduced by 4.33%
session	ID	20298	1649.11	12.31	<.0001 → residual variance not reduced

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8297.7	16	8329.7	8330.7	8346.7	8371.6	8387.6

Is the reasoning model (5a) better than the age model (4a)?

Yes, $-2\Delta LL = 12.5$ on $df=3$, $p=.0059$, so ΔR^2 is significant

Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept	1966.47	49.6658	101	39.59	<.0001	
c1sess	-119.74	19.7742	101	-6.06	<.0001	
c1sess*c1sess	13.3036	3.3656	101	3.95	0.0001	
age80	22.2782	8.6018	101	2.59	0.0110	
c1sess*age80	-6.4921	3.4247	101	-1.90	0.0609	
c1sess*c1sess*age80	0.9601	0.5829	101	1.65	0.1026	
reas22	-27.1004	11.1141	101	-2.44	0.0165	
c1sess*reas22	-3.5917	4.4250	101	-0.81	0.4189	
c1sess*c1sess*reas22	1.1575	0.7531	101	1.54	0.1274	

Pearson Correlation Coefficients, N = 606

Prob > |r| under H0: Rho=0

	nm3rt	Pred
nm3rt	1.00000	0.40108
nm3rt: Number-Match 3 RT in ms		<.0001

$R=.4011$, so R^2 for time+age+reas = .1609

The fixed effects of time and age before accounted for ~10.7% of the variance in RT, so there is a net increase of ~5.4% due to reasoning.

Model 5b. Abstract Reasoning on Intercept and Linear Time Slope Only

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + U_{2i}$$

```

TITLE "SAS Model 5b: Reasoning on Intercept and Linear Time Slope Only";
PROC MIXED DATA=work.example23 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
              reas22 clsess*reas22
              / SOLUTION DDFM=Satterthwaite OUTPM=work.ReasPred2;
  RANDOM INTERCEPT clsess clsess*clsess / G GCORR TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  * Requesting additional effects for reasoning instead;
  ESTIMATE "Reasoning Effect at Session 1" reas22 1 clsess*reas22 0;
  ESTIMATE "Reasoning Effect at Session 2" reas22 1 clsess*reas22 1;
  ESTIMATE "Reasoning Effect at Session 3" reas22 1 clsess*reas22 2;
  ESTIMATE "Reasoning Effect at Session 4" reas22 1 clsess*reas22 3;
  ESTIMATE "Reasoning Effect at Session 5" reas22 1 clsess*reas22 4;
  ESTIMATE "Reasoning Effect at Session 6" reas22 1 clsess*reas22 5;
RUN; PROC CORR NOSIMPLE DATA=work.ReasPred2; VAR nm3rt pred; RUN;

TITLE "SPSS Model 5b: +Reasoning as Predictor of Intercept, Linear Time Slope Only".
MIXED nm3rt BY ID session WITH clsess age80 reas22
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
          reas22 clsess*reas22
  /RANDOM = INTERCEPT clsess clsess*clsess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (predreas2)
  /TEST = "Reasoning Effect at Session 1" reas22 1 clsess*reas22 0
  /TEST = "Reasoning Effect at Session 2" reas22 1 clsess*reas22 1
  /TEST = "Reasoning Effect at Session 3" reas22 1 clsess*reas22 2
  /TEST = "Reasoning Effect at Session 4" reas22 1 clsess*reas22 3
  /TEST = "Reasoning Effect at Session 5" reas22 1 clsess*reas22 4
  /TEST = "Reasoning Effect at Session 6" reas22 1 clsess*reas22 5.
CORRELATIONS nm3rt predreas2.

* STATA Model 5b: +Reasoning as Predictor of Intercept, Linear Time Slope Only
xtmixed nm3rt c.clsess c.clsess#c.clsess ///
  c.age80 c.age80#c.clsess c.age80#c.clsess#c.clsess ///
  c.reas22 c.reas22#c.clsess, || id: clsess clsess2, ///
  variance ml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store Reas2, // save LL for LRT
  lrtest Reas2 Age, // LRT against age baseline
  margins, at(c.clsess=(0(1)5)) dydx(c.reas22) vsquish // reas slope per session
  margins, at(c.clsess=(0(1)5) c.reas22=(-5 0 5)) vsquish // predictions per session
  marginsplot, name(predicted_reas, replace) // plot reas predictions
  predict predreas2 // save fixed-effect predicted outcomes
corr nm3rt predreas2 // get total r to make r2

```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	228688	34635	6.60	<.0001 → intercept variance reduced by 5.68%
UN(2,1)	ID	-31959	10877	-2.94	0.0033
UN(2,2)	ID	24872	5711.58	4.35	<.0001 → linear slope variance reduced by -2.38%
UN(3,1)	ID	3877.66	1786.86	2.17	0.0300
UN(3,2)	ID	-3766.27	957.50	-3.93	<.0001
UN(3,3)	ID	606.16	167.70	3.61	0.0002 → quad slope variance reduced by 0.03%
session	ID	20298	1649.11	12.31	<.0001 → residual variance not reduced

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8300.1	15	8330.1	8330.9	8345.9	8369.3	8384.3

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1969.80	49.6821	101	39.65	<.0001
c1sess	-123.54	19.8277	101	-6.23	<.0001
c1sess*c1sess	13.9774	3.3754	101	4.14	<.0001
age80	20.8470	8.5613	103	2.44	0.0166
c1sess*age80	-4.8610	3.2905	102	-1.48	0.1427
c1sess*c1sess*age80	0.6709	0.5580	101	1.20	0.2321
reas22	-32.8284	10.4706	101	-3.14	0.0022
c1sess*reas22	2.9363	1.2412	101	2.37	0.0199

Is the revised reasoning model (5b) still better than the age model (4a)?

Yes, $-2\Delta LL = 10.1$ on $df=2$, $p=.006$ (so only 2.4 of the previous $-2\Delta LL$ was due to $reason*quad$)

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > t	
Reasoning Effect at Session 1	-32.8284	10.4706	101	-3.14	0.0022	
Reasoning Effect at Session 2	-29.8921	9.9615	101	-3.00	0.0034	
Reasoning Effect at Session 3	-26.9558	9.5870	101	-2.81	0.0059	
Reasoning Effect at Session 4	-24.0195	9.3632	101	-2.57	0.0118	
Reasoning Effect at Session 5	-21.0831	9.3012	101	-2.27	0.0255	
Reasoning Effect at Session 6	-18.1468	9.4040	101	-1.93	0.0564	

These are the simple slopes for the effect of reasoning per session.

Pearson Correlation Coefficients, N = 606


Prob > |r| under H0: Rho=0

	nm3rt	Pred
nm3rt	1.00000	0.40008
nm3rt		<.0001

$R = .4001$, so R^2 for $time+age+reas = .1601$

So ~0.1% of the variance accounted for previously was due to $reason*quad$

This plot was produced by the marginsplot in STATA.



Model 6a. +Education Group on Intercept, Linear, and Quadratic Time Slopes

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + \gamma_{03}(\text{Highvs.LowEd}_i) + \gamma_{04}(\text{Highvs.MedEd}_i) + U_{0i}$$

$$\text{Linear: } \beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + \gamma_{13}(\text{Highvs.LowEd}_i) + \gamma_{14}(\text{Highvs.MedEd}_i) + U_{1i}$$

$$\text{Quadratic: } \beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{23}(\text{Highvs.LowEd}_i) + \gamma_{24}(\text{Highvs.MedEd}_i) + U_{2i}$$

Additional model-implied group differences:

$$\text{Medium vs. Low education intercept} = (\gamma_{00} + \gamma_{04}) - (\gamma_{00} + \gamma_{03}) = \gamma_{04} - \gamma_{03}$$

$$\text{Medium vs. Low education linear session} = (\gamma_{10} + \gamma_{14}) - (\gamma_{10} + \gamma_{13}) = \gamma_{14} - \gamma_{13}$$

$$\text{Medium vs. Low education quadratic session} = (\gamma_{20} + \gamma_{24}) - (\gamma_{20} + \gamma_{23}) = \gamma_{24} - \gamma_{23}$$

```
TITLE "SAS Model 6a: +Education Group on Intercept, Linear, and Quadratic";
PROC MIXED DATA=work.example23 NOCLPRINT NOITPRINT COVTEST NAMELEN=100 IC METHOD=ML;
  CLASS ID session educgrp;
  MODEL nm3rt = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
               reas22 clsess*reas22 educgrp clsess*educgrp clsess*clsess*educgrp
               / SOLUTION DDFM=Satterthwaite OUTPM=work.EducPred;
  RANDOM INTERCEPT clsess clsess*clsess / G GCORR TYPE=UN SUBJECT=ID;
  REPEATED session / TYPE=VC SUBJECT=ID;
  * Estimating group means at first and last sessions
  LSMEANS educgrp / AT (clsess) = (0) DIFF=ALL;
  LSMEANS educgrp / AT (clsess) = (5) DIFF=ALL;
  * Contrasts between groups on intercept, linear, and quadratic slopes
  ESTIMATE "L vs. H Educ for Intercept Main Effect" educgrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Intercept Main Effect" educgrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Intercept Main Effect" educgrp -1 1 0 ;
  ESTIMATE "L vs. H Educ for Linear Session" clsess*educgrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Linear Session" clsess*educgrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Linear Session" clsess*educgrp -1 1 0 ;
  ESTIMATE "L vs. H Educ for Quadratic Session" clsess*clsess*educgrp -1 0 1 ;
  ESTIMATE "M vs. H Educ for Quadratic Session" clsess*clsess*educgrp 0 -1 1 ;
  ESTIMATE "L vs. M Educ for Quadratic Session" clsess*clsess*educgrp -1 1 0 ;
RUN; PROC CORR NOSIMPLE DATA=work.EducPred; VAR nm3rt pred; RUN;
```

Think of the -1 as the "0" and the "1" as the "1" in a dummy code.

The table of model fixed effects will have the highest group as the 0 instead, though.

```
TITLE "SPSS Model 6a: +Education as Predictor of Intercept, Linear, and Quadratic".
MIXED nm3rt BY ID session educgrp WITH clsess age80 reas22
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV G R
  /FIXED = clsess clsess*clsess age80 clsess*age80 clsess*clsess*age80
           reas22 clsess*reas22 educgrp clsess*educgrp clsess*clsess*educgrp
  /RANDOM = INTERCEPT clsess clsess*clsess | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = session | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (prededuc)
  /EMMEANS = TABLES(educgrp) WITH (clsess=0) COMPARE(educgrp)
  /EMMEANS = TABLES(educgrp) WITH (clsess=5) COMPARE(educgrp)
  /TEST = "L vs. H Educ for for Main Effect" educgrp -1 0 1
  /TEST = "M vs. H Educ for for Main Effect" educgrp 0 -1 1
  /TEST = "L vs. M Educ for for Main Effect" educgrp -1 1 0
  /TEST = "L vs. H Educ for for Linear Session" clsess*educgrp -1 0 1
  /TEST = "M vs. H Educ for for Linear Session" clsess*educgrp 0 -1 1
  /TEST = "L vs. M Educ for for Linear Session" clsess*educgrp -1 1 0
  /TEST = "L vs. H Educ for for Quadratic Session" clsess*clsess*educgrp -1 0 1
  /TEST = "M vs. H Educ for for Quadratic Session" clsess*clsess*educgrp 0 -1 1
  /TEST = "L vs. M Educ for for Quadratic Session" clsess*clsess*educgrp -1 1 0.
CORRELATIONS nm3rt prededuc.
```

```

* STATA Model 6a: +Education Group on Intercept, Linear, and Quadratic
xtmixed nm3rt c.clsess c.clsess#c.clsess ///
  c.age80 c.age80#c.clsess c.age80#c.clsess#c.clsess ///
  c.reas22 c.reas22#c.clsess ///
  ib(last).educgrp ib(last).educgrp#c.clsess ///
  ib(last).educgrp#c.clsess#c.clsess, || id: clsess clsess2, ///
  variance ml covariance(un) residuals(independent,t(session)),
  estat ic, n(101),
  estat recovariance, level(id),
  estimates store Educ,
  lrtest Educ Reas2,
* Estimating group means at first and last sessions
  margins ib(last).educgrp, at(c.clsess=(0 5))
* Contrasts between groups on intercept, linear, and quadratic slopes
  test 1.educgrp=3.educgrp // Low vs. High: Intercept
  test 2.educgrp=3.educgrp // Med vs. High: Intercept
  test 1.educgrp=2.educgrp // Low vs. Med: Intercept
  test 1.educgrp#c.clsess=3.educgrp#c.clsess // Low vs. High: Linear
  test 2.educgrp#c.clsess=3.educgrp#c.clsess // Med vs. High: Linear
  test 1.educgrp#c.clsess=2.educgrp#c.clsess // Low vs. Med: Linear
  test 1.educgrp#c.clsess#c.clsess=3.educgrp#c.clsess#c.clsess // Low vs. High: Quad
  test 2.educgrp#c.clsess#c.clsess=3.educgrp#c.clsess#c.clsess // Med vs. High: Quad
  test 1.educgrp#c.clsess#c.clsess=2.educgrp#c.clsess#c.clsess // Low vs. Med: Quad
  contrast educgrp, // omnibus group diff on intercept
  contrast educgrp#c.clsess, // omnibus group diff on linear
  contrast educgrp#c.clsess#c.clsess, // omnibus group diff on quadratic
  margins, at(c.clsess=(0(1)5) educgrp=(1 2 3)) vsquish // predictions per session
  marginsplot, name(predicted_educ, replace) // plot educ predictions
  predict prededuc // save fixed-effect predicted outcomes
corr nm3rt prededuc // get total r to make r2

```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	228585	34693	6.59	<.0001 → intercept variance reduced by 0.05%
UN(2,1)	ID	-33273	10909	-3.05	0.0023
UN(2,2)	ID	24129	5614.80	4.30	<.0001 → linear slope variance reduced by 2.99%
UN(3,1)	ID	4125.93	1788.44	2.31	0.0211
UN(3,2)	ID	-3633.38	939.20	-3.87	0.0001
UN(3,3)	ID	581.50	164.31	3.54	0.0002 → quad slope variance reduced by 4.07%
session	ID	20298	1649.11	12.31	<.0001 → residual variance not reduced

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8295.4	21	8337.4	8338.9	8359.6	8392.3	8413.3

Is the education model (6a) better than the revised reasoning model (5b)?

No, $-2\Delta LL = 4.7$ on $df=6$, $p = .583$

Solution for Fixed Effects						
Education						
Effect	Group (1=HS, 2=BA, 3=GRAD)	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		1961.89	101.79	100	19.27	<.0001
clsess		-106.50	40.2761	101	-2.64	0.0095
clsess*clsess		12.4797	6.8474	101	1.82	0.0713
age80		20.2893	8.5600	102	2.37	0.0196
clsess*age80		-4.5758	3.2667	102	-1.40	0.1643
clsess*clsess*age80		0.6177	0.5533	101	1.12	0.2669
reas22		-36.6228	10.7638	101	-3.40	0.0010
clsess*reas22		2.9788	1.2799	101	2.33	0.0219
educgrp	1	-51.3811	151.06	101	-0.34	0.7345
educgrp	2	37.6427	120.87	100	0.31	0.7561
educgrp	3	0

c1sess*educgrp	1	-70.2445	59.0672	101	-1.19	0.2371
c1sess*educgrp	2	-4.3577	48.1238	100	-0.09	0.9280
c1sess*educgrp	3	0
c1sess*c1sess*educgrp	1	11.0653	10.0300	101	1.10	0.2726
c1sess*c1sess*educgrp	2	-1.4641	8.1865	101	-0.18	0.8584
c1sess*c1sess*educgrp	3	0

Type 3 Tests of Fixed Effects (truncated)

Effect	Num DF	Den DF	F Value	Pr > F
educgrp	2	101	0.24	0.7874
c1sess*educgrp	2	101	0.96	0.3860
c1sess*c1sess*educgrp	2	101	1.09	0.3395

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
L vs. H Educ for Intercept Main Effect	51.3811	151.06	101	0.34	0.7345
M vs. H Educ for Intercept Main Effect	-37.6427	120.87	100	-0.31	0.7561
L vs. M Educ for Intercept Main Effect	89.0238	130.74	102	0.68	0.4975
L vs. H Educ for Linear Session	70.2445	59.0672	101	1.19	0.2371
M vs. H Educ for Linear Session	4.3577	48.1238	100	0.09	0.9280
L vs. M Educ for Linear Session	65.8868	50.7047	101	1.30	0.1967
L vs. H Educ for Quadratic Session	-11.0653	10.0300	101	-1.10	0.2726
M vs. H Educ for Quadratic Session	1.4641	8.1865	101	0.18	0.8584
L vs. M Educ for Quadratic Session	-12.5294	8.6028	101	-1.46	0.1484

Least Squares Means

Effect	Education Group (1=HS, 2=BA,3=GRAD)	c1sess	age80	reas22	Estimate	Standard Error	DF	t Value	Pr > t
educgrp	1	0.00	-0.17	0.62	1884.28	110.93	101	16.99	<.0001
educgrp	2	0.00	-0.17	0.62	1973.30	66.5665	100	29.64	<.0001
educgrp	3	0.00	-0.17	0.62	1935.66	101.24	100	19.12	<.0001
educgrp	1	5.00	-0.17	0.62	1599.71	94.5405	101	16.92	<.0001
educgrp	2	5.00	-0.17	0.62	1704.94	56.6069	101	30.12	<.0001
educgrp	3	5.00	-0.17	0.62	1725.69	86.0606	101	20.05	<.0001

Differences of Least Squares Means

Effect	Education Group (1=HS, 2=BA, 3=GRAD)	Education Group (1=HS, 2=BA, 3=GRAD)	c1sess	age80	reas22	Estimate	Standard Error	DF	t Value	Pr > t
educgrp	1	2	0.00	-0.17	0.62	-89.0238	130.74	102	-0.68	0.4975
educgrp	1	3	0.00	-0.17	0.62	-51.3811	151.06	101	-0.34	0.7345
educgrp	2	3	0.00	-0.17	0.62	37.6427	120.87	100	0.31	0.7561
educgrp	1	2	5.00	-0.17	0.62	-105.22	111.48	101	-0.94	0.3475
educgrp	1	3	5.00	-0.17	0.62	-125.97	128.68	101	-0.98	0.3299
educgrp	2	3	5.00	-0.17	0.62	-20.7487	102.72	101	-0.20	0.8403

Pearson Correlation Coefficients, N = 606

Prob > |r| under H0: Rho=0

	nm3rt	Pred
nm3rt	1.00000	0.41510
nm3rt: Number-Match 3 RT in ms		<.0001

R =.41510, so R^2 for time+age+reas+educ = .172

The fixed effects of time, age, and reasoning before accounted for ~16.0% of the variance in RT, so there is a net increase of 1.2% due to education (which is not significant).

Simple Processing Speed: Example Conditional Models of Change Results

The extent to which individual differences in response time (RT) over six sessions for a simple processing speed test (number match three) could be predicted from baseline age, abstract reasoning, and education level was examined in a series of multilevel models (i.e., general linear mixed models) in which the six practice sessions were nested within each participant. Given the interest in comparing models differing in fixed effects, maximum likelihood (ML) was used in estimating and reporting all model parameters; denominator degrees of freedom were estimated using the Satterthwaite method. The significance of new fixed effects were evaluated with individual Wald tests (i.e., of estimate / SE) as well as with likelihood ratio tests (i.e., $-2\Delta LL$), with degrees of freedom equal to the number of new fixed effects. Session (i.e., the index of time) was centered at the first occasion, age was centered at 80 years, abstract reasoning was centered at 22 (near the mean of the scale), and graduate-level education was the reference group for education level (with separate contrasts for high school or less and for bachelor's level education).

The best-fitting unconditional growth model specified quadratic decline across the six sessions (i.e., a decelerating negative function) with significant individual differences in the intercept, linear, and quadratic effects. Accordingly, effect size was evaluated via pseudo- R^2 values for the proportion reduction in each random effect variance, as well as with total R^2 , the squared correlation between the actual outcome values and the outcomes predicted by the model fixed effects. In the unconditional growth model, the fixed effects for linear and quadratic change across sessions accounted for approximately 4% of the total variation in RT.

Next, age was added as a predictor of the intercept, linear slope, and quadratic slope. The age model fit significantly better than the unconditional model as indicated by a significant likelihood ratio test, $-2\Delta LL(3) = 11.6$, $p = .009$; the AIC was lower, although the BIC was not. However, only the fixed effect of age on the intercept was significant, indicating that for every additional year of age above 80, RT at the first session was predicted to be significantly higher by 29.05 ($p = .0008$). In terms of pseudo- R^2 , age accounted for 11.29% of the random intercept variance, 4.50% of the random linear slope variance, and 2.64% of the random quadratic slope variance. As expected given that baseline age is a time-invariant predictor, the residual variance was not reduced. The total cumulative R^2 from session and age was $R^2 = .11$, approximately a 7% increase due to age (which was significant, as indicated by the likelihood ratio test). Although the interactions of age with the linear and quadratic slopes were not significant, they were retained in the model to fully control for age effects before examining the effects of other predictors.

Abstract reasoning was then added as a predictor of the intercept, linear slope, and quadratic slope. The abstract reasoning model fit significantly better than the age model, $-2\Delta LL(3) = 12.5$, $p = .006$; the AIC was lower, although the BIC was not. However, only the fixed effect of reasoning on the intercept was significant. The nonsignificant effect of reasoning on the quadratic slope was then removed, revealing a significant effect of reasoning on both the intercept and linear slope, such that for every unit higher reasoning above 22, RT at the first session was expected to be lower by 32.82 and the linear rate of improvement in RT (as evaluated at the first session given the quadratic slope) was expected to be less negative by 2.94 (i.e., faster initial RT with less improvement in persons with greater reasoning). These two effects still resulted in a significant improvement in model fit over the age model, $-2\Delta LL(2) = 10.1$, $p = .006$, with a lower AIC and BIC. Reasoning accounted for 5.58% of the random intercept variance but had no measurable reduction of the random linear and quadratic slope variances. The total cumulative R^2 from session, age, and reasoning was $R^2 = .16$, approximately a 5% increase due to reasoning (which was significant, as indicated by the likelihood ratio test).

Finally, education level (high school or less, bachelor's level, or graduate level) was then added as a predictor of the intercept, linear slope, and quadratic slope. The education model did not fit significantly better than the reasoning model, $-2\Delta LL(6) = 4.7$, $p = .583$, with a higher AIC and BIC. None of the omnibus main effects of group on the intercept, linear, or quadratic slopes were significant, $F's(2,101) < 1.10$, $p's > .05$, and none of the pairwise group comparisons were significant as well. Education accounted for 0.05% of the random intercept variance, 2.99% of the random linear slope variance, and 4.07% of the random quadratic slope variance. The total cumulative R^2 from session, age, reasoning, and education was $R^2 = .17$, approximately a 1% increase due to education (which was not significant, as indicated by the likelihood ratio test). (From here one might remove nonsignificant model effects and/or add other effects as needed to fully answer all research questions...)